Занятие 2. Условные операторы. Переменные с плавающей точкой и операции над ними. Логический тип. Интервальный тип.

С помощью всего того, что мы изучили на занятии 1, мы уже можем написать множество программ, делающих множество различных вещей. Вспомним, однако, как сложно нам было даже написать программу, находящую максимум из двух чисел. Наша проблема заключалась в том, что какие бы входные данные не поступали на вход нашей программы, она все равно выполняла одни и те же действия (хоть и для разных входных данных). Было бы очень хорошо, если бы программа могла принимать решения в зависимости от входных данных. Например, если A<B, то совершать какие-нибудь одни действия, а в противном случае – какие-нибудь другие. В этом случае, очевидно, задача нахождения максимума стала бы просто очевидной. Итак, сегодня мы будем учить нашу программу принимать решения в зависимости от входных данных.

Этой цели можно добиться с помощью использования условных операторов. Но сначала мы познакомимся с новым типом переменных. Это нам поможет лучше понять работу условных операторов. Это – логический тип (Boolean). Переменные этого типа могут принимать 2 значения: истина (True) и ложь (False). То есть, если некоторая переменная Flag описана следующим образом:

var


Flag: Boolean;

то допустимы присваивания «Flag:= True;» и «Flag:= False;».

Когда мы используем какую-либо логическую переменную в программе, всегда полезно представлять, что эта переменная является результатом вычисления некоторого логического выражения, значение которого зависит от входных данных. Снова вернемся к примеру с нахождением максимума из двух чисел. В данном случае можно рассмотреть в качестве логического выражения «A<B» и использовать в программе некоторую логическую переменную, которая будет принимать значение True, если A<B, и значение «False», если A≥B.
Отметим, что каждая переменная логического типа занимает один байт в памяти.
Рассмотрим некоторые операции, которые применимы к переменным логического типа:
1) Операция «and» (логическое и). Данная операция применяется к двум переменным логического типа (или к выражениям, результат вычисления которых имеет логический тип) и результат ее применения имеет логический тип. Данная операция работает следующим образом:
	A
	B
	A and B

	True 
	True
	True

	True
	False
	False

	False
	True
	False

	False
	False
	False


Чтобы понять смысл операции «and», предположим, что переменные A и B – результаты вычисления некоторых логических выражений a и b. Тогда мы можем рассмотреть выражение c = «a и b». Понятно, что это выражение будет истинным в тех и только тех случаях, когда оба выражения a и b истинны. Поэтому, если мы присвоим переменной C значение A and B, то переменная C будет содержать значение выражения c.
2) Операция «or» (логическое или). Данная операция применяется к двум переменным логического типа (или к выражениям, результат вычисления которых имеет логический тип) и результат ее применения имеет логический тип. Данная операция работает следующим образом:

	A
	B
	A or B

	True 
	True
	True

	True
	False
	True

	False
	True
	True

	False
	False
	False


Чтобы понять смысл операции «and», предположим, что переменные A и B – результаты вычисления некоторых логических выражений a и b. Тогда мы можем рассмотреть выражение c = «a или b». Понятно, что это выражение будет истинным в тех и только тех случаях, когда хотя бы одно из выражений a и b истинно. Поэтому, если мы присвоим переменной C значение A or B, то переменная C будет содержать значение выражения c.
3) Операция «xor» (исключающее или). Данная операция применяется к двум переменным логического типа (или к выражениям, результат вычисления которых имеет логический тип) и результат ее применения имеет логический тип. Данная операция работает следующим образом:

	A
	B
	A xor B

	True 
	True
	False

	True
	False
	True

	False
	True
	True

	False
	False
	False


Чтобы понять смысл операции «xor», предположим, что переменной A соответствует лампочка, которая или включена (A = True), или выключена (A = False). Переменной B поставим в соответствие выключатель, который исправен (B = True), или неисправен (B = False). Тогда, если C = A xor B, то C = True, если лампочка будет включена после нажатия на переключатель, и C = false в противном случае.

Еще одна интерпретация операции xor получается, если рассмотреть представление логических переменных в памяти компьютера. Дело в том, что в памяти любая переменная хранится как число. В частности, значение True обычно хранится как 1, а значение False, как 0. Тогда, как нетрудно проверить, операцию A xor B можно интерпретировать в числах как (A+B) mod 2 (сложение по модулю 2).
Хотя операция xor кажется немного лишней, в дальнейшем мы увидим много интересных примеров ее применения.

4) Операция «not» (отрицание). Данная операция применяется к одной переменной логического типа (или к выражению, результат вычисления которого имеет логический тип) и результат ее применения имеет логический тип. Данная операция работает следующим образом:
	A
	not A

	true
	false

	false
	true


Чтобы понять смысл операции «not», предположим, что переменная A – результат вычисления логического выражения a. Рассмотрим выражение b = «не a». Тогда присваивание B:= not A помещает в переменную B значение выражения b.
Как и в случае переменных целого типа, мы можем составлять с логическими переменными при помощи определенных выше операций различные выражения. При этом для указания порядка действий лучше использовать скобки (второй вариант заключается в том, чтобы заглянуть в справочник по языку Паскаль и узнать приоритеты для вышеприведенных операций).

Чтобы лучше запомнить логические операции, приведем несколько тождеств для этих операций:

1) A and A = A; A or A = A; A xor A = False

2) A and B = B and A; A or B = B or A; A xor B = B xor A

3) A and (B and C) = (A and B) and C; A or (B or C) = (A or B) or C; A xor (B xor C) = (A xor B) xor C
4) A and (not A) = False; A or (not A) = True; A xor (not A) = True

5) A and (B or C) = (A and B) or (A and C)

6) A or (B and C) = (A or B) and (A or C)

7) A and (B xor C) = (A and B) xor (A and C)
8) not (A and B) = (not A) or (not B)

9) not (A or B) = (not A) and (not B)
Не составляет труда придумать еще множество подобных тождеств. Все они могут быть доказаны простой проверкой всех возможных вариантов.
Данные тождества можно использовать для упрощения логических выражений.
И снова вернемся к примеру с нахождением максимума из двух чисел. Как говорилось выше, мы можем рассмотреть некоторую логическую переменную с целью хранения значения вычисления выражения «A<B». Но как мы сможем поместить в эту переменную это значение? По-видимому, никакие хитрые операции and, or, not или xor нам здесь не помогут. Однако, Паскаль располагает набором операций, которые применяются к целым числам, но результат их применения имеет логический тип. Рассмотрим некоторые из этих операций:
1) Операция «<» (меньше). Данная операция применяется к двум переменным (или выражениям) числового типа (как мы увидим дальше, не только числового) и результат ее применения имеет логический тип. После присваивания «C:= A<B;» (переменная C – логического типа!!!) переменная C получит значение True, если A<B, и значение False в противном случае (то есть, если A≥B). Как раз это присваивание и вычисляет значение выражения «A<B». Остальные операции имеют абсолютно аналогичный формат, поэтому приведем их без подробного описания.
2) Операция «>» (больше).
3) Операция «<=» (меньше, либо равно).
4) Операция «>=» (больше, либо равно).
5) Операция «=» (равно).
6) Операция «<>» (не равно).
Понятно, что мы можем использовать логические операции и операции сравнения в одном и том же логическом выражении. Рассмотрим пример составления достаточно сложного логического выражения.
Пример. Рассмотрим фигуру «крест», изображенную ниже.

[image: image1]
Пусть крест имеет ширину 12 и высоту 10 и симметричен относительно точки (0, 0). Напишем логическое выражение, которое в зависимости от координат некоторой точки (X, Y) получает значение «истина», если эта точка лежит в кресте, и значение «ложь» в противном случае.
Будем составлять выражение по частям. Сначала составим выражение «точка (X, Y) лежит в горизонтальной полосе креста». Понятно, что точка (X, Y) лежит в горизонтальной полосе тогда, и только тогда, когда X лежит в отрезке [-5; 5], то есть, когда X≤5 и X≥-5. Поэтому нужное нам логическое выражение имеет вид (X<=5) and (X>=-5). Аналогично выражение «точка (X, Y) лежит в вертикальной полосе креста» имеет вид (Y<=6) and (Y>=-6). Наконец, для составления нужного нам выражения, осталось заметить, что точка лежит внутри креста тогда и только тогда, когда она лежит или в горизонтальной, или в вертикальной полосе. Поэтому искомое выражение может быть записано в виде ((X<=5) and (X>=-5)) or ((Y<=6) and (Y>=-6)).
Понятно, что это выражение можно было бы составить и другими способами. Например, с использованием свойства 6) логических операций получаем:
((X<=5) and (X>=-5)) or ((Y<=6) and (Y>=-6)) = (((X<=5) and (X>=-5)) or (Y<=6)) and (((X<=5) and (X>=-5)) or (Y>=-6)) = ((X<=5) or (Y<=6)) and ((X>=-5) or (Y<=6)) and ((X<=5) or (Y>=-6)) and ((X>=-5) or (Y>=-6)) – еще два варианта составления того же самого выражения (хоть и более сложных). Полезно проверить наши действия, изобразив последнее полученное выражение на координатной плоскости. Например, выражение (X<=5) or (Y<=6) вырезает правый верхний угол координатной плоскости. Аналогичным образом действуют остальные три выражения, вырезая оставшиеся три угла. Применение операции and к этим четырем выражениям позволяет вырезать сразу все четыре угла и получить требуемый нам крест.
Теперь, после того, как мы получили некоторое представление о логических переменных, можно перейти к обучению программ принимать решения. Для этого нам нужно изучить условные оператор языка Паскаль.

Сначала мы рассмотрим оператор if (если). В самом простом варианте его структура имеет вид:

if логическое выражение then
  оператор;
Под логическим выражением понимается любое выражение, результат вычисления которого имеет логический тип. Под оператором понимается любой оператор языка Паскаль (в том числе и еще один оператор if). Оператор if работает следующим образом: вычисляется логическое выражение; если результат вычисления этого выражения – истина (true), то выполняется оператор, в противном случае ничего не происходит (действие переходит к следующему оператору после if). Слово then переводится с английского как «то». Таким образом, мы имеем структуру вида «если что-то верно, то сделай что-то». Например, мы можем написать:
if A<B then

  WriteLn(A, ‘ меньше ‘, B);

Вопрос: что делать, если мы хотим вместо одного оператора выполнить несколько? Ответ: использовать составной оператор «begin..end». Мы немного коснулись этого оператора, когда рассматривали структуру программы. Рассмотрим его подробнее. Оператор «begin..end» имеет следующую структуру:

begin

  оператор1;

  …

  операторn;
end;

Этот оператор ничего не делает. Это просто контейнер для произвольных других операторов. То есть, при выполнении составного оператора выполняются по очереди оператор1, оператор2, …, операторn. Таким образом, назначение составного оператора состоит в том, чтобы объединять много операторов в один.

Запишем структуру оператора «if» для выполнения нескольких операторов:

if логическое выражение then begin

  оператор1;

  …

  операторn;
end;

Сделаем два замечания. Во-первых, эта структура ничем не отличается от той, которая была написана в самом начале. Просто мы вместо оператора подставили конкретный составной оператор. Во-вторых, точка с запятой после оператораn не обязательна.

Вообще, было бы полезно иметь такой оператор, который при выполнении некоторого условия передает управление одной группе операторов, а при его невыполнении – другой. Этого также можно добиться с использованием оператора if с измененной структурой. Эта структура приведена ниже:
if логическое выражение then

  оператор1

  else
  оператор2;

В данном варианте оператор if работает следующим образом: вычисляется логическое выражение; если оно истинно (true), то выполняется оператор1, иначе – оператор2; после этого (независимо от значения логического выражения) управление передается следующему после if оператору. Отметим, что оператор1 и оператор2 – произвольные операторы, в том числе и составные. Кроме того, точку с запятой после оператора1 ставить не нужно.

Пример. В качестве примера применения оператора if решим задачу «Найдите максимум» с предыдущего занятия. Теперь решение выглядит намного проще.

program FindMax;

var

  A, B: LongInt;

begin

  ReadLn(A, B);

  If A>B then

    WriteLn(A)

    else

    WriteLn(B);

end.

Кроме того, что программа стала несравнимо проще и понятнее, она работает теперь для любых A и B в интервале [-231; 231-1]. Вот какого прогресса мы смогли добиться!!!

Следующий условный оператор мы рассмотрим на примере. Предположим, что нам нужно ввести с экрана число A (0≤A≤9) и напечатать его словами (например, если A = 1, то напечатать «один»). Понятно, что эту задачу несложно решить с помощью использования оператора if. Программа приведена ниже:
program WriteDigit;

var

  A: LongInt;

begin

  ReadLn(A);

  if A = 0 then WriteLn(‘ноль’);

  if A = 1 then WriteLn(‘один’);

  if A = 2 then WriteLn(‘два’);

  if A = 3 then WriteLn(‘три’);

  if A = 4 then WriteLn(‘четыре’);

  if A = 5 then WriteLn(‘пять’);

  if A = 6 then WriteLn(‘шесть’);

  if A = 7 then WriteLn(‘семь’);

  if A = 8 then WriteLn(‘восемь’);

  if A = 9 then WriteLn(‘девять’);

end.

Однако, программа очень сильно нагружена операторами if. Оказывается, вместо 10 операторов if можно использовать всего 1 оператор множественного ветвления case.

Структура этого оператора такова:

case выражение of

  вариант1: оператор1;

  вариант2: оператор2;

  …

  вариантn: операторn
  else оператор

end;
Под «выражением» здесь понимается абсолютно любое выражение, но с единственным ограничением: результат его вычисления должен иметь перечислимый тип. Проще говоря, количество возможных значений, принимаемых этим типом, должно быть конечным. В частности, целые и логические типы являются перечислимыми.

Вместо «операторов» можно, как обычно, произвольные операторы, в том числе и составные.

«Варианты» описывают некоторое множество значений того типа, который имеет результат вычисления выражения. Каждый «вариант» имеет следующую структуру:

1) Конкретное значение, которое может принимать тип выражения, является вариантом. Например, если выражение имеет тип LongInt, то в качестве варианта можно использовать любое число из интервала [-231; 231-1].
2) Интервал значений, которые может принимать тип выражения, является вариантом. Снова предположим, что выражение имеет тип LongInt. В данном случае мы можем использовать любой подинтервал [a, b] интервала [-231; 231-1]. При этом для задания такого интервала [a, b] в качестве варианта оператора case, используется запись «a..b».
3) Объединение нескольких интервалов и/или конкретных значений, которые может принимать тип выражения, является вариантом. При этом для задания такого объединения в качестве варианта оператора case каждый интервал и/или конкретное значение задается, так как описано выше, и все интервалы и/или конкретные значения отделяются друг от друга запятыми. Например, если выражение имеет тип LongInt, то вполне допустим следующий вариант: «0, 5, 10..15». Он задает множество значений {0, 5, 10, 11, 12, 13, 14, 15}.

Отметим также, что каждое множество, задаваемое вариантами, должно содержать хотя бы одно значение. Кроме того, все эти множества должны попарно не пересекаться, то есть не содержать общих значений. Наконец, часть «else оператор» оператора case необязательна.
Оператор case работает следующим образом: сначала вычисляется значение выражения; далее ищется такой «вариантi», который задает множество, содержащее значение выражения (такой вариант, если он есть, единственен, так как множества, задаваемые вариантами, попарно не пересекаются); если такой вариант найден, то выполняется «операторi»; если такого варианта нет, то выполняется «оператор» (который стоит после else; если такого оператора нет, то ничего не выполняется); после всего этого действие переходит к следующему после case оператору.
Рассмотрим несколько примеров.

Пример 1. Решим еще раз рассмотренную выше задачу: ввести число A (0≤A≤9) и напечатать его словами. Решение имеет следующий вид:

program WriteDigit;

var

  A: LongInt;

begin

  ReadLn(A);

  case A of

    0: WriteLn(‘ноль’);
    1: WriteLn(‘один’);
    2: WriteLn(‘два’);
    3: WriteLn(‘три’);
    4: WriteLn(‘четыре’);
    5: WriteLn(‘пять’);
    6: WriteLn(‘шесть’);
    7: WriteLn(‘семь’);
    8: WriteLn(‘восемь’);
    9: WriteLn(‘девять’)
  end;

end.
Пример 2. Написать программу, которая вводит число A (0≤A≤9) и выводит, является оно четным или нечетным. Решение имеет следующий вид:
program OddOrEven;

var

  A: LongInt;

begin

  ReadLn(A);

  case A of

    0, 2, 4, 6, 8: WriteLn(‘четное’)

    else WriteLn(‘нечетное’)

  end;
end.

Пример 3. Написать программу, которая вводит число A (0≤A≤999) и выводит, является оно однозначным, двузначным или трехзначным. Решение имеет следующий вид:

program HowMuchDigits;

begin

  ReadLn(A);

  case A of

    0..9: WriteLn(‘однозначное’);

    10..99: WriteLn(‘двузначное’)

    else WriteLn(‘трехзначное’)

  end;
end.

Рассмотрим еще один момент. Понятно, что понимается под интервалом значений некоторого типа, если этот тип – целый. Но уже в случае, когда тип – логический, это неочевидно. Чтобы пояснить это, достаточно вспомнить, что каждая переменная, независимо от ее типа, хранится в памяти как число. Поэтому, на самом деле, нет принципиальной разницы между целыми типами и всеми остальными. Более точно, интервал [a, b] в случае, когда a и b – переменные нецелых типов, понимается следующим образом: пусть переменная a хранится в памяти как число A, переменная B – как число b, тогда в интервал [a, b] попадают все те значения, которые хранятся в памяти как числа из интервала [A, B]. Например, в интервал [false, true] попадают значения false и true, а в интервал [true, false] не попадает ни одного значения.

Рассмотрим теперь еще одну возможность языка Паскаль – интервальные типы. Возможность эта состоит в том, что мы в качестве типа можем указать произвольный интервал произвольного перечислимого типа. Например, вполне допустима запись:
var

  A: 0..10000;

  B: false..true;

Преимущество использования таких типов заключается в упрощении контроля правильности программы. Например, Вы знаете, что некоторая переменная A во время работы вашей программы может принимать только значения от 0 до 10. Предположим, однако, что программа написана с ошибкой, и в некоторый момент она приняла значение 20. Кроме того, предположим, что Вы включили в опциях компилятора проверку переполнения. Тогда, если переменная A описана как переменная типа Byte, то программа нормально отработает, и Вы можете даже не заметить ошибку. Если же переменная A описана как переменная типа 0..10, то, как только она в процессе программы она примет значение 20, сразу возникнет сообщение об ошибке. При этом Вы поймете, что где-то в программе есть ошибка, найдете ее и исправите.
Для того, чтобы определить, какое количество памяти занимает переменная интервального типа, необходимо отыскать перечислимый тип, который занимает минимальное количество памяти и полностью покрывает все значения данного интервального типа. Тогда переменная интервального типа будет занимать столько же памяти, сколько и переменная найденного перечислимого типа. Так, переменные типа 250..255 занимают 1 байт памяти, переменные типа 256..257 – 2 байта, а переменные типа 2000000000..2100000000 – 4 байта.

Перейдем теперь к последней части нашего занятия – описанию переменных вещественных типов (или, как их еще называют – переменных с плавающей точкой). Необходимость таких переменных достаточно очевидна – числа бывают не только целыми, а и дробными. Язык Паскаль предоставляет пять типов для работы с вещественными числами. Эти типы различаются:
1) Способом реализации. Большинство компьютеров содержат специальное устройство для быстрого выполнения различных сложных арифметических вычислений – арифметический сопроцессор. В частности, с помощью сопроцессора можно выполнять арифметические действия над вещественными числами – это один из способов реализации вещественных типов. Однако, раньше не все компьютеры имели сопроцессоры. Если на компьютере нет сопроцессора, то все, что остается компилятору – это самому моделировать действия над вещественными числами. Это – второй из способов реализации вещественных типов. Естественно, что использование второго способа дает меньшую скорость выполнения действий.
2) Количеством занимаемой памяти.
3) Диапазон принимаемых значений.
4) Количество значимых цифр. Вещественные типы имеют очень большой диапазон значений, но, естественно, никакое вещественное значение не может храниться абсолютно точно. Действительно, как мы можем хранить точно число 

, если оно имеет бесконечно много цифр после запятой и не имеет периода?!? На самом деле, хранится только определенное количество значимых цифр (например, 10 самых левых цифр), а точность значения остальных цифр гарантировать нельзя (чаще всего они заменяются на ноли).
Таблица ниже характеризует вещественные типы языка Паскаль:

	Название типа
	Способ реализации
	Количество занимаемой памяти
	Диапазон принимаемых значений
	Количество значимых цифр

	Real
	Моделирование
	6

[-1040; 1040]
	11-12

	Single
	Сопроцессор
	4
	
[image: image4.wmf]»

[-1040; 1040]
	7-8

	Double
	Сопроцессор
	8

[-10300; 10300]
	15-16

	Extended
	Сопроцессор
	10

[-104900; 104900]
	19-20

	Comp
	Сопроцессор
	8

[-1018; 1018]
	19-20


Как видно из таблицы, у каждого из типов есть свои плюсы и свои минусы. Тип Single является самым неточным, но он занимает мало памяти. Более точным является тип Real, но он занимает уже больше памяти и не использует сопроцессор. Поэтому использовать тип Real стоит только, если скорость работы вашей программы не очень критична. Еще более точным является тип Double, и снова растет количество занимаемой памяти. Наконец, самым точным является тип Extended, который, однако, занимает целых 10 байт памяти. Именно этот тип нужно использовать, для хорошей точности вычислений, если у Вас есть достаточное количество памяти.
Отдельно от этих типов стоит тип Comp. Он относится к вещественным типам, так как над ним допустимы те же самые операции, что и над вещественными типами. Однако в этом типе хранятся только целые значения. То есть, когда некоторое вещественное число присваивается переменной типа Comp, то все знаки после запятой у этого числа просто теряются (число не округляется, а просто исчезают все знаки после запятой). В этом смысле, тип Comp можно считать целочисленным. Тип Comp является неплохим типом, когда нам нужно работать с достаточно большими целыми числами (порядка 14-15 знаков). При этом следует иметь в виду, что этот тип не является точным, то есть не любое число из диапазона принимаемых значений хранится точно. Поэтому не стоит использовать этот тип для работы с числами порядка 17-18 знаков, так как результат вычислений в этом случае вполне может оказаться неточным.

Над вещественными числами, в отличие от целых, применим немного другой набор операций. Операции +, -, * применимы и над вещественными числами. А вот операции div, mod применимы только к целым числам. Вместо них у вещественных чисел появляется операция деления /. При этом, если мы присваиваем результат деления переменной типа Comp, то оно даже будет целочисленным.
Рассмотрим еще несколько операций, применимых к числовым типам (через несколько занятий мы скажем, что это на самом деле не операции, а функции, но пока будем называть их операциями).

1) Abs(числовой тип) – модуль числа. Эта операция применяется к одному числу любого типа (как целого, так и вещественного). В результате ее применения получается величина такого же типа, как и число, к которому она была применена. Эта операция позволяет находить модуль числа X, т. е. число 

2) ArcTan(числовой тип) – арктангенс числа. Эта операция применяется к одному числу любого типа и результат ее применения имеет вещественный тип. Эта операция позволяет находить арктангенс числа X, то есть восстанавливать угол по его тангенсу. Угол, полученный в результате, будет выражен в радианах и принадлежать интервалу 

3) Cos(числовой тип) – косинус угла. Эта операция применяется к числу любого типа и результат ее применения имеет вещественный тип. Эта операция позволяет находить косинус угла, выраженного в радианах.

4) Exp(числовой тип) – экспонента числа. Эта операция применяется к числу любого типа и результат ее применения имеет вещественный тип. Эта операция позволяет по числу X находит eX, где e = 2.7172… - основание натуральных логарифмов.
5) Frac(числовой тип) – дробная часть числа. Эта операция применяется к числу любого типа и результат ее применения имеет вещественный тип. Эта операция находит дробную часть числа X. Для целых чисел X выполняется тождество Frac(X) = 0. Если X < 0, то Frac(X) = -Frac(-X). Таким образом, Frac(5.3) = 0.3; Frac(-4.7) = -0.7.

6) Int(числовой тип) – целая часть числа. Эта операция применяется к числу любого типа и результат ее применения имеет вещественный тип. Эта операция находит целую часть числа X. Для целых чисел X выполняется тождество Int(X) = X (на самом деле это не совсем корректно, так как X – целое число, а Int(X) – вещественное). Если X < 0, то Int(X) = -Int(-X). Таким образом, Int(8.9) = 8; Int(-10.3) = -10.
7) Ln(числовой тип) – натуральный логарифм числа. Эта операция применяется к числу любого типа и результат ее применения имеет вещественный тип. Эта операция находит по числу X число lnX – натуральный логарифм числа X. Для корректной работы этой операции должно выполняться условие X>0.

8) Pi – число «пи». Эта операция ни к чему не применяется, она просто возвращает вещественное значение, приблизительно равное значению числа 

 - отношения длины окружности к диаметру.
9) Sin(числовой тип) – синус угла. Эта операция применяется к числу любого типа и результат ее применения имеет вещественный тип. Эта операция позволяет находить синус угла, выраженного в радианах.

10)  Sqr(числовой тип) – квадрат числа. Эта операция применяется к числу любого типа и результат ее применения имеет тот же тип, что и аргумент. Она позволяет по числу X находить X2. Зачем нужна такая операция? Почему бы не написать X*X? Есть две причины. Первая: X может быть не только числом, но и выражением, в том числе и очень длинным. В этом случае запись X*X содержит почти в два раза больше символов, чем запись Sqr(X). Вторая: операция Sqr может работать быстрее (на некоторых компиляторах), чем умножение, так как возведение в квадрат является достаточно частным случаем умножения.
11)  Sqrt(числовой тип) – квадратный корень из числа. Эта операция применяется к числу любого типа и результат ее применения имеет вещественный тип. Эта операция позволяет по числу X находить значение 

. Для корректной ее работы должно выполняться условие X

12)  Round(числовой тип) – округление числа до целых. Эта операция применяется к числу любого типа и результат ее применения имеет целый тип. Она округляет число X до ближайшего целого. Если число X лежит ровно посередине между двумя соседними целыми числами, то округление ведется в сторону числа с большим модулем. Так, Round(7.3) = 7; Round(-5.7) = -6; Round(5.5) = 6; Round(-5.5) = -6.
13) Trunc(числовой тип) – целая часть числа (еще одна). Эта операция применяется к числу любого типа и результат ее применения имеет целый тип. Во всем остальном она работает так же, как и операция Int.
Разберем еще один вопрос, связанный с выводом вещественных чисел на экран. Если Вы напишите следующую программу:
program WriteFloat;

var

  x: Extended;

begin

  x:= 0.5;

  WriteLn(x);

end.
и запустите ее, то на экране увидите не 0.5, а нечто вроде «5.00000000000000E-0001». В общем-то, эта безобидная надпись означает всего лишь 

, но неплохо было бы научиться выводить вещественные числа в более читаемом виде. Для этого мы можем выводить вещественные числа с отформатированном виде. Для этого, при выводе числа X напишем после него два числа A и B, разделенные двоеточием, то есть выведем число X в виде «X:A:B». Число A указывает, какое общее количество символов нужно потратить на вывод числа X (если этих знаков окажется слишком много, то в начале числа будет напечатано некоторое количество пробелов), число B указывает, какое число знаков после запятой у числа X нужно вывести. Если Вы укажете A = 0, то будет выведено ровно столько знаков, сколько занимает число X. Если Вы укажете B = 0, то число будет выведено до целых. Обычно, если в задаче требуется выводить вещественные числа, то указывается, сколько знаков после запятой нужно вывести у ответа.
Еще одна тонкость использования вещественных чисел заключается в том, что, как уже говорилось, все вычисления с вещественными числами проводятся неточно. Это означает, что все операции сравнения с использованием вещественных чисел нужно проводить аккуратно. Предположим, что Вы вычислили некоторые выражения с участием вещественных чисел и поместили их значения в переменные a и b. Пусть далее, если a=b, то вы хотите выполнить некоторый один набор действий, а в противном случае – некоторый другой. Логично, что Вы будете использовать конструкцию «If a=b then … else …». И можете очень легко на этом залететь!!! Причина очень проста – при разработке алгоритма Вы исходите из предположения абсолютно точных вычислений. При разработке программы Вы должны учитывать неточность вычислений. Если a=b при условии абсолютно точных вычислений, то это совершенно не обязательно так, когда вычисления неточны. Вы вполне можете очутиться в ситуации, когда из-за погрешностей вычислений переменные a и b получат очень близкие (очень-очень близкие), но все-таки не равные значения, хотя при абсолютной точности вычислений они должны были бы получить равные значения. В результате Ваша программа выполнит совсем другой набор действий и отработает неправильно. Во избежание таких случаев сравнение вещественных чисел обычно проводят с некоторой точностью. В частности, вместо условия a=b обычно используется условие Abs(a-b)<e, где e – точность сравнения. Обычно e полагают достаточно малым, например, e = 0.00000001. В этом случае Ваша программа окажется гораздо более защищенной от погрешностей вычислений.
В заключение рассмотрим еще один пример.

Пример. Треугольник задан координатами своих вершин (X1, Y1), (X2, Y2), (X3, Y3). Числа Xi, Yi – целые и не превышают 1000 по модулю. Необходимо вычислить площадь треугольника S и вывести ее на экран с точностью до 3-х знаков после запятой.

Решение. Мы рассмотрим самый плохой способ решения этой задачи. Из математики известна следующая формула Герона нахождения площади треугольника по его сторонам a, b, c: S = 
[image: image14.wmf])

)(

)(

(

c

p

b

p

a

p

p

-

-

-

, где p = (a+b+c)/2 – полупериметр треугольника. Для нахождения сторон a, b, c нужно научиться считать расстояние между двумя точками на плоскости. Для этого можно использовать теорему Пифагора.

Если точка A имеет координаты (x1, y1), а B – координаты (x2, y2), то AB = |x1-x2|, а BC = |y1-y2|. Так как треугольника ABC – прямоугольный, то AC2 = AB2+BC2. Таким образом, AC = 

. Вот и все. Программа приведена ниже:

program TriangleMeasure;

var

  x1, y1, x2, y2, x3, y3: LongInt;

  a, b, c, p: Extended;

  s: Extended;

begin

  ReadLn(x1, y1);

  ReadLn(x2, y2);

  ReadLn(x3, y3);

  a:= Sqrt(Sqr(x1-x2)+Sqr(y1-y2));

  b:= Sqrt(Sqr(x1-x3)+Sqr(y1-y3));

  c:= Sqrt(Sqr(x2-x3)+Sqr(y2-y3));

  p:= (a+b+c)/2;

  s:= Sqrt(p*(p-a)*(p-b)*(p-c));

  WriteLn(s:0:3);

end.

Проясним еще один вопрос – почему этот способ решения плохой? Причины, как всегда, две. Во-первых, данная программа работает медленно. Это объясняется тем, что операция извлечения квадратного корня является очень медленной. Конечно, если мы запустим программу один раз, то нам покажется, что она работает быстро. Но если нам понадобится находить площади сразу большого количества треугольниклв, то может понадобиться более эффективный способ. Во-вторых, эта программа работает не очень точно. Причина состоит, опять же, в том, что операция извлечения квадратного корня является неточной.

